

# NAG Fortran Library Routine Document

## F07HDF (SPBTRF/DPBTRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07HDF (SPBTRF/DPBTRF) computes the Cholesky factorization of a real symmetric positive-definite band matrix.

### 2 Specification

```
SUBROUTINE F07HDF (UPLO, N, KD, AB, LDAB, INFO)
ENTRY      spbtrf (UPLO, N, KD, AB, LDAB, INFO)
INTEGER   N, KD, LDAB, INFO
real     AB(LDAB,*)
CHARACTER*1 UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine forms the Cholesky factorization of a real symmetric positive-definite band matrix  $A$  either as  $A = U^T U$  if UPLO = 'U', or  $A = LL^T$  if UPLO = 'L', where  $U$  (or  $L$ ) is an upper (or lower) triangular band matrix with the same number of super-diagonals (or sub-diagonals) as  $A$ .

### 4 References

Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: UPLO – CHARACTER\*1 *Input*

*On entry:* indicates whether the upper or lower triangular part of  $A$  is stored and how  $A$  is factorized, as follows:

if UPLO = 'U', the upper triangular part of  $A$  is stored and  $A$  is factorized as  $U^T U$ , where  $U$  is upper triangular;

if UPLO = 'L', the lower triangular part of  $A$  is stored and  $A$  is factorized as  $LL^T$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

- 3: KD – INTEGER *Input*  
*On entry:*  $k$ , the number of super-diagonals or sub-diagonals of the matrix  $A$ .  
*Constraint:*  $KD \geq 0$ .
- 4: AB(LDAB,\*) – *real* array *Input/Output*  
**Note:** the second dimension of the array AB must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  symmetric band matrix  $A$ , stored in rows 1 to  $k+1$ . More precisely, if UPLO = 'U', the elements of the upper triangle of  $A$  within the band must be stored with element  $a_{ij}$  in  $AB(k+1+i-j, j)$  for  $\max(1, j-k) \leq i \leq j$ ; if UPLO = 'L', the elements of the lower triangle of  $A$  within the band must be stored with element  $a_{ij}$  in  $AB(1+i-j, j)$  for  $j \leq i \leq \min(n, j+k)$ .  
*On exit:* the upper or lower triangle of  $A$  is overwritten by the Cholesky factor  $U$  or  $L$  as specified by UPLO, using the same storage format as described above.
- 5: LDAB – INTEGER *Input*  
*On entry:* the first dimension of the array AB as declared in the (sub)program from which F07HDF (SPBTRF/DPBTRF) is called.  
*Constraint:*  $LDAB \geq KD + 1$ .
- 6: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the leading minor of order  $i$  is not positive-definite and the factorization could not be completed. Hence  $A$  itself is not positive-definite. This may indicate an error in forming the matrix  $A$ . There is no routine specifically designed to factorize a symmetric band matrix which is not positive-definite; the matrix must be treated either as a nonsymmetric band matrix, by calling F07BDF (SGBTRF/DGBTRF) or as a full symmetric matrix, by calling F07MDF (SSYTRF/DSYTRF).

## 7 Accuracy

If UPLO = 'U', the computed factor  $U$  is the exact factor of a perturbed matrix  $A + E$ , where

$$|E| \leq c(k+1)\epsilon|U^T||U|,$$

$c(k+1)$  is a modest linear function of  $k+1$ , and  $\epsilon$  is the *machine precision*.

If UPLO = 'L', a similar statement holds for the computed factor  $L$ . It follows that  $|e_{ij}| \leq c(k+1)\epsilon\sqrt{a_{ii}a_{jj}}$ .

## 8 Further Comments

The total number of floating-point operations is approximately  $n(k+1)^2$ , assuming  $n \gg k$ .

A call to this routine may be followed by calls to the routines:

F07HEF (SPBTRS/DPBTRS) to solve  $AX = B$ ;

F07HGF (SPBCON/DPBCON) to estimate the condition number of  $A$ .

The complex analogue of this routine is F07HRF (CPBTRF/ZPBTRF).

## 9 Example

To compute the Cholesky factorization of the matrix  $A$ , where

$$A = \begin{pmatrix} 5.49 & 2.68 & 0.00 & 0.00 \\ 2.68 & 5.63 & -2.39 & 0.00 \\ 0.00 & -2.39 & 2.60 & -2.22 \\ 0.00 & 0.00 & -2.22 & 5.17 \end{pmatrix}.$$

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07HDF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, KMAX, LDAB
PARAMETER       (NMAX=8,KMAX=8,LDAB=KMAX+1)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, KD, N
CHARACTER       UPLO
*      .. Local Arrays ..
real           AB(LDAB,NMAX)
*      .. External Subroutines ..
EXTERNAL        spbtrf, X04CEF
*      .. Intrinsic Functions ..
INTRINSIC       MAX, MIN
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07HDF Example Program Results'
Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, KD
IF (N.LE.NMAX .AND. KD.LE.KMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        DO 20 I = 1, N
          READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
20      CONTINUE
      ELSE IF (UPLO.EQ.'L') THEN
        DO 40 I = 1, N
          READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
40      CONTINUE
      END IF
*
*      Factorize A
*
      CALL spbtrf(UPLO,N,KD,AB,LDAB,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Print factor
*
        IFAIL = 0

```

```

*
      IF (UPLO.EQ.'U') THEN
*
          CALL X04CEF(N,N,0,KD,AB,LDAB,'Factor',IFAIL)
*
      ELSE IF (UPLO.EQ.'L') THEN
*
          CALL X04CEF(N,N,KD,0,AB,LDAB,'Factor',IFAIL)
*
      END IF
*
      ELSE
          WRITE (NOUT,*) 'A is not positive-definite'
      END IF
      END IF
      STOP
*
      END

```

## 9.2 Program Data

F07HDF Example Program Data

```

4 1          :Values of N and KD
'L'         :Value of UPLO
5.49
2.68  5.63
      -2.39  2.60
          -2.22  5.17  :End of matrix A

```

## 9.3 Program Results

F07HDF Example Program Results

Factor	1	2	3	4
1	2.3431			
2	1.1438	2.0789		
3		-1.1497	1.1306	
4			-1.9635	1.1465

---